Digitalizovaná kapitola skript M.Malíka – digitalizace dosud neprošla korekturou

15. TŘÍDA SIMULATION.

Třída simulation je třetí systémovou standardní třídou, která obsahuje prostředky problémově orientovaného subjazyka určené pro diskrétní simulaci.

Metodami diskrétní simulace rozumíme metody sloužící k simulaci systémů, které se mění v konečném časovém intervalu pouze v konečně mnoha časových okamžicích (nebo o nichž lze takovou zákonitost vývoje předpokládat).

Jednou z možných metod diskrétní simulace - na které je též založen princip systémové třídy simulation - je metoda tak zvaného plánování událostí.

Simulační výpočty pak probíhají v krocích odpovídajících okamžikům změn simulovaného systému, přičemž v každém tomto okamžiku jsou naplánovány další kroky a naopak eventuelně zrušeny některé z plánů učiněných v předcházejících krocích.

((15.01.01.)), ((15.01.02.))

15.01. DEFINICE TŘÍDY SIMULATION.

Syntakticky lze definici třídy simulation popsat následovně:


simset class simulation;


begin



link class EVENTNOTICE ~~~;



link class process ~~~;



ref (head) SQS;



ref (EVENTNOTICE) procedure FIRSTEV ~~~;



ref (process) procedure current ~~~;



real procedure time ~~~;



procedure hold ~~~;



procedure passivate ~~~;



procedure wait ~~~;



procedure cancel ~~~;



procedure ACTIVATE ~~~;



procedure accum ~~~;



process class MAINPROGRAM ~~~;



ref (MAINPROGRAM) main;



SQS :- new head;



main :- new MAINPROGRAM;



main.EVENT :- new EVENTNOTICE (0, main);



main.EVENT.into (SQS)


end
Systémová třída simulation je tedy podtřídou systémové třídy simset.

Třída EVENTNOTICE slouží k plánování simulačních akcí, které jsou popsány v jednotlivých podtřídách třídy process - přesněji v jejich operačních částech.

Spojový seznám SQS představuje posloupnost plánů simulačních akcí (to jest exemplářů třídy EVENTNOTICE) a vždy bude uspořádán podle hodnot simulárního času daného plánu.

Spojový seznam SQS nazýváme kalendářem událostí.

15.02. KALENDÁŘE UDÁLOSTÍ.

15.02.01. Provádění simulačního procesu.

Jak bylo již výše uvedeno, provádí se simulační výpočet ve třídě simulation na základě kvaziparalelního systému postupným předáváním řízení jednotlivým exemplářům třídy process a hlavním programem prefixovaného bloku.

Prostředky kvaziparalelního systému jsou uživateli skryty a uživatel pouze plánuje výpočet podle jednotlivých exemplářů třídy process (přesněji podtřídy třídy process) v určitý okamžik simulárního času.

Simulární čas je zde fiktivní - během výpočtu neklesající - veličina, která odpovídá skutečnému času při vývoji simulovaného systému.

15.02.02. Pojem kalendáře událostí.

Kalendářem událostí můžeme tedy rozumět prostředek, který registruje uživatelem zadané plány výpočtu jednotlivých exemplářů podtříd třídy process a v každém simulačním kroku vybírá z těchto plánů ten, který má nejmenší hodnotu simulárního času.

Přirozeným požadavkem na kalendář událostí je, aby do něj bylo možno další plány přidávat a naopak z něj dříve dříve zařazené plány vypouštět.

15.02.03. Realizace spojovým seznamem.

Jednou z možných realizací kalendáře událostí je realizace spojovým seznamem, v němž jsou plány provádění jednotlivých akcí seřazeny podle vzrůstajících hodnot simulárního času.

V syntaktické definici třídy simulation odpovídá této možnosti realizace exemplář třídy head označený uživateli nepřístupnou referenční proměnnou SQS.

15.03. TŘÍDA EVENTNOTICE.

Pomocí exemplářů třídy EVENTNOTICE jsou realizovány plány simulačních akcí, které jsou zařazovány do kalendáře SQS.

15.03.01. Syntaktická definice.

Syntaktickou definici třídy EVENTNOTICE je možno podle normy jazyka popsat následovně:


link class EVENTNOTICE (EVTIME, PROC);


real EVTIME;


ref (process) PROC;


begin


ref (EVENTNOTICE) procedure suc;



suc :- if SUC in EVENTNOTICE then SUC else none;



ref (EVENTNOTICE) procedure pred;



pred :- PRED;



procedure RANK (BEFORE);



boolean BEFORE;



begin



ref (EVENTNOTICE) P;




P :- SQS.last;




for P :- P while P.EVTIME gt EVTIME do



P :- P.pred;




if BEFORE then



begin




for P :- P while P.EVTIME eq EVTIME do




P :- P.pred




end;




follow (P)



end;


end
15.03.02. Parametry.

Parametr PROC exempláře třídy EVENTNOTICE odkazuje na exemplář třídy process, kterého se daná položka kalendáře událostí týká.

Parametr EVTIME téhož exempláře obsahuje reálný číselný údaj, který se interpretuje jako okamžik simulárního času, ve kterém má být aktivován exemplář třídy process daný parametrem PROC (neboli okamžik, kdy má být tomuto exempláři třídy process předáno řízení).

15.03.03. Procedury suc a pred.

Procedury suc a pred předeklarovávají stejnojmenné procedury deklarované v prototypu třídy linkage.

Tyto referenční funkční procedury nejsou kvalifikovány pro třídu link, ale přímo pro třídu EVENTNOTICE. Jejich deklarace je přizpůsobena použití v jiných procedurách třídy simulation - jejich přímé použití uživatelem není možné.

15.03.04. Procedura rank.

Procedura RANK provádí zařazení příslušného exempláře třídy EVENTNOTICE do spojového seznamu SQS podle hodnoty jeho atributu EVTIME. (Předpokládáme, že seznam SQS je uspořádán podle hodnot atributů EVTIME od nejmenší hodnoty k největší.)

Je-li parametr procedury BEFORE roven při vyvolání hodnotě false, provede se zařazení za všechny exempláře v seznamu SQS se stejnou hodnotou atributu EVTIME, je-li naopak skutečný parametr BEFORE roven konstantě true, provede se zařazení před všechny takové exempláře.

Upozorněme, že předpokládáme, že nově zařazovaný exemplář třídy EVENTNOTICE nebude mít menší hodnotu parametru EVTIME než prvý prvek seznamu SQS.

15.04. PROCEDURY CURRENT A TIME.

Poznamenejme předem, že prototyp třídy process obsahuje referenční atribut EVENT kvalifikovaný pro třídu EVENTNOTICE, který u každého exempláře třídy process odkazuje na element seznamu SQS, jehož parametr PROC zpět na uvažovaný exemplář třídy process. Neexistuje-li takový element kalendáře událostí, je hodnotou atributu EVENT uvažovaného exempláře třídy process konstanta none.

15.04.01. Procedura firstev.

Referenční funkční procedura FIRSTEV vydává prvý plán zařazený v seznamu SQS.

Tento element seznamu SQS odkazuje vždy svým parametrem PROC na exemplář třídy process, který je právě aktivní.


ref (EVENTNOTICE) procedure FIRSTEV;


FIRSTEV :- SQS.first;

l5.04.02. Procedura current.

Referenční funkční procedura current vydává vždy exemplář třídy process, jehož plán je první položkou kalendáře událostí SQS.

Jak bylo výše uvedeno, vyplývá z filosofie třídy simulation, že tento exemplář třídy process musí být právě aktivní v kvaziparalelním systému, ve kterém simulační výpočet probíhá.


ref (process) procedure current;


current :- FIRSTEV.PROC;

((15.02.02.))

15.04.03. Simulární čas a procedura time.

Již bylo výše uvedeno, že simulárním časem rozumíme nezápornou veličinu typu real, jejíž hodnoty se postupně a průběhem výpočtu zvětšují, a která představuje průběh skutečného času při vývoji skutečného simulovaného systému.

Ve třídě simulation není simulární čas realizován samostatnou reálnou proměnnou, za jeho hodnotu se totiž považuje hodnota parametru EVTIME prvého plánu zařazeného v kalendáři událostí SQS.

Tato hodnota je zpřístupněna reálnou funkční procedurou time.


real procedure time;


time := FIRSTEV.EVTIME;

((15.02.03.))

15.05. TŘÍDA PROCESS.

Exempláře třídy process tedy představují prototypy simulačních událostí, které jsou prováděny v jistých okamžicích simulárního času.

Aby nedošlo k porušení vazeb mezi realizujícím kvaziparalelním systémem a datovými strukturami třídy simulation, nedoporučuje se uživateli používat v částech programu prefixovaných třídou simulation standardní procedury pro řízení kvaziparalelního systému, totiž procedury detach, resume a call.

15.05.01. Syntaktická definice.

Podle normy jazyka je možno syntaktickou definici třídy process zapsat takto:


link class process;


begin


ref (EVENTNOTICE) EVENT;



boolean TERMINATED;



boolean procedure idle;



idle := EVENT == none;



boolean procedure terminated;



terminated := TERMINATED;



real procedure evtime;



if idle then ERROR



else


evtime := EVENT.EVTIME;



ref (process) procedure nextev;



nextev :-



if idle then none


else


if EVENT.suc == none then none


else


EVENT.suc.PROC;



detach;



inner;



TERMINATED := true;



passivate;



ERROR


end
15.05.02. Stavy exempláře třídy process.

Podle vztahu k realizujícímu kvaziparalelnímu systému a k řídícím datovým strukturám třídy simulation se může každý exemplář třídy process nacházet v jednom ze čtyř stavů.

Tyto stavy jsou:


aktivní


suspendovaný


pasivní


ukončený

((15.03.01.)), ((15.03.02.)), ((15.03.03.))

15.05.02.01. Aktivní stav.

V aktivním stavu je vždy právě jeden exemplář třídy process, který je též aktivní ve smyslu realizujícího kvaziparalelního systému.

Tento exemplář má svůj plán události zařazen na začátku kalendáře SQS a je zpřístupněn procedurou current.

15.05.02.02. Suspendovaný stav.

V suspendovaném stavu je takový exemplář třídy process, který má svůj plán události zařazen v kalendáři událostí SQS, ale ne na prvním místě.

V rámci realizujícího kvaziparalelního systému je takový exemplář neaktivní a samostatný.

15.05.02.03. Pasivní stav.

V pasivním stavu je takový exemplář třídy process, který nemá vůbec naplánováno své aktivování, to znamená, že jeho atribut EVENT je roven referenční konstantě none a že v kalendáři událostí SQS není žádný element, který by na tento exemplář třídy process odkazoval.

Přitom je exemplář třídy process, který je v pasivním stravu, v rámci realizujícího kvaziparalelního systému neukončený, neaktivní a samostatný.

15.05.02.04. Ukončený stav.

Konečně v ukončeném stavu je takový exemplář třídy process, který již vyčerpal svou operační část a jehož atribut TERMINATED je roven konstantě true.

Na rozdíl od syntaktické definice podle normy jazyka je toto většinou realizováno tak, že exemplář třídy process v ukončeném stavu je též ukončen ve smyslu realizujícího kvaziparalelního systému.

15.05.03. Procedura idle.

Logická funkční procedura idle vydává hodnotu true právě tehdy, je-li příslušný exemplář třídy process ve stavu pasivním nebo ukončeném.

((15.03.04.))

15.05.04. Procedura terminated.

Podobně vydává logická funkční procedura terminated hodnotu true právě v tom případě, když příslušný exemplář třídy process je ve stavu ukončeném.

((15.05.04.))

15.05.05. Procedura evtime.

V případě, že exemplář třídy process je ve stavu aktivním nebo suspendovaném, vydá jeho reálná funkční procedura evtime hodnotu času plánu jeho aktivace (v případě exempláře ve stavu aktivním je to právě hodnota simulárního času vydávaná též procedurou time).

V ostatních případech stavů exempláře třídy process dojde při volání procedury evtime k chybě při výpočtu.

((15.03.04.))

15.05.06. Procedura nextev.

Referenční funkční procedura nextev vydává odkaz na exemplář třídy process, jehož provedení je naplánováno právě po provedení daného exempláře.

Je-li běžný exemplář třídy process ve stavu pasivním nebo ukončeném nebo je-li plán provedení běžného exempláře poslední položkou kalendáře událostí, vydá procedura nextev referenční konstantu none.

((15.03.05.))

15.05.07. Operační část.

Prvým příkazem operační části prototypu třídy process je volání standardní procedury detach. Tím se každý exemplář třídy process stane ihned při vygenerování samostatným ve smyslu řídícího kvaziparalelního systému a je v pasivním stavu.

((15.03.06.))

Znamená to mimo jiné, že při generování exempláře podtřídy třídy process se vůbec nezačnou provádět příkazy operační části deklarace prototypu této podtřídy.

Teprve po aktivaci takového exempláře se provádí příkazy operační části definované uživatelem (příkaz inner) - během provádění této části se příslušný exemplář může opakovaně dostat do stavu aktivního, suspendovaného i pasivního.

Po ukončení uživatelské části se změní hodnota atributu TERMINATED a exemplář se stane ukončeným. Volání standardní procedury passivate (viz dále) jej učiní neaktivním ve smyslu kvaziparalelního systému. Další pokus o předání řízení tomuto prvku vede k chybě při výpočtu.

((15.03.07.))

15.06. HLAVNÍ PROGRAM.

15.06.01. Prefixovaný blok třídy simulation.

V kvaziparalelním systému, ve kterém probíhá simulační výpočet může být aktivní nejenom některý z exemplářů třídy process, ale i dynamická representace samotného hlavního prefixovaného bloku.

Protože exempláře třídy process nejsou řízeny přímo prostředky pro řízení kvaziparalelního systému, ale zprostředkovaně pomocí řídících datových struktur třídy simulation, je účelné zajistit, aby s dynamickou reprezentací hlavního prefixovaného bloku bylo možno zacházet ve smyslu simulačního plánování výpočtu jako s jedním exemplářem třídy process (přesněji jako s exemplářem jisté podtřídy třídy process).

15.06.02. Třída mainprogram.

Pro generování exempláře třídy process, který představuje hlavní prefixovaný blok slouží třída MAINPROGRAM, podtřída třídy process, která je uživateli nepřístupná.

Její syntaktická definice je podle normy jazyka následující:


process class MAINPROGRAM;


begin


m:



detach;



go to m


end
15.06.03. Proměnná main.

V operační části třídy simulation je vytvořen exemplář třídy MAINPROGRAM a je pro uživatele zpřístupněn referenční proměnnou main.

Hlavní prefixovaný blok bude tedy v rámci řídícího kvaziparalelního systému aktivní právě tehdy, když plán provádění exempláře main bude prvý na časové ose simulačního kalendáře (t.j. prvním elementem seznamu SQS).

15.06.04. Operační část třídy simulation.

Vyjma příkazů výše uvedených se v operační části třídy simulation vytvoří plán provádění exempláře main a zařadí se s časem nula do kalendáře událostí.

Znamená to, že v okamžiku začátku výpočtu prefixovaného bloku třídy simulation (resp. některé její podtřídy) je exemplář třídy MAINPROGRAM representující dynamický exemplář tohoto prefixovaného bloku ve stavu aktivním a zůstává ve stavu aktivním až do té doby, kdy změnou kalendáře událostí přestane být plán jeho provádění první.

((15.04.01.)), ((15.04.02.)), ((15.04.03.))

15.06.05. Ukončení simulačního výpočtu.

Přesně podle zásad výpočtu kvaziparalelního systému končí simulační výpočet právě tehdy, když přejde provádění příkazů hlavního prefixovaného bloku přes příslušné koncové end.

Nechceme-li, aby simulační výpočet skončil v simulárním čase nula, musíme změnit plánování aktivace exempláře main (například použitím procedury passivate nebo hold - viz dále - v operační části prefixovaného bloku nebo v operační části podtřídy třídy simulation).

((15.04.04.)), ((15.04.05.))

15.07. PASIVUJÍCÍ PŘÍKAZY.

Příkazy řídící provádění simulačního výpočtu lze rozdělit na pasivující a aktivující.

Prvou skupinou jsou ty, které mění stav procesu z aktivního do suspendovaného či pasivního nebo ze stavu suspendovaného do stavu pasivního.

Všechny tyto změny stavů procesů se provádějí prostřednictvím změn příslušných plánů aktivace v kalendáři událostí. Pasivující příkazy vyjímají plány aktivace z kalendáře nebo je posouvají vzad ve směru časové osy, aktivující příkazy naopak vkládají plány aktivace na příslušné místo v kalendáři událostí.

15.07.01. Procedura hold.

Standardní procedura hold je podle normy jazyka popsána následující deklarací:


procedure hold(t);


real t;


inspect FIRSTEV do

begin


if t gt 0 then


EVTIME := EVTIME + t;



if suc =/= none then


begin



if suc.EVTIME le EVTIME then



begin




out;





RANK (false);





resume (current)




end


end

end
Použití procedury hold s kladným reálným parametrem t způsobí, že plán provádění právě aktivního procesu zvětší svůj plán aktivace (přesněji časovou složku svého plánu aktivace) o hodnotu parametru t. Tedy k další aktivaci dosud aktivního exempláře třídy process dojde v momentě, až bude hodnota simulárního času o t větší než v okamžiku volání procedury hold. (Samozřejmě, nebude-li mezi tím změněn kalendář událostí.)

Je-li v kalendáři událostí plán provedení jiného procesu s časem menším nebo rovným nové hodnotě času plánu dosud aktivního procesu, stane se dosud dosud aktivní proces suspendovaným a řízení se předá procesu, jehož plán aktivace je nyní první v kalendáři událostí.

Je-li procedura hold volána s nekladným parametrem, způsobí takové volání plánu provedení dosud aktivního procesu za všechny plány zařazené v kalendáři událostí se stejným časem (existují-li nějaké).

((15.05.01.))

Schematicky ukazuje význam procedury hold následující obrázek.

15.07.02. Procedura passivate.

Deklarace procedury passivate je podle normy jazyka následující:


procedure passivate;


begin


inspect current do


begin



EVENT.out;




EVENT :- none


end;



if SQS.empty then ERROR



else


resume (current)


end
Volání procedury passivate změní stav dosud aktivního exempláře třídy process na pasivní - vyřadí totiž jeho plán aktivace, který byl dosud na prvém místě, z kalendáře.

V případě, že v kalendáři událostí není žádný další plán aktivace, dojde k chybě při výpočtu, jinak se zaktivuje proces, jehož plán provedení je na prvém místě v kalendáři SQS.

((15.03.03.))

Schematicky je opět činnost procedury passivate znázorněna na následujícím obrázku.

15.07.03. Procedura wait.


procedure wait (s);


ref (head) s;


begin


current.into (s);



passivate


end
Standardní procedura wait v podstatě rozšiřuje činnost procedury passivate.

Před vlastním provedením procedury passivate se dosud aktivní exemplář třídy process zařadí na konec spojového seznamu, který je dán skutečným parametrem procedury wait.

(Poznamenejme, že třída process je podtřídou třídy link, takže její exempláře mohou být elementy spojových seznamů tvořených exempláři třídy head.)

15.07.04. Procedura cancel.

Na rozdíl od předcházejících tří procedur, které mění stav dosud dosud aktivního procesu, může procedura cancel měnit stav libovolného suspendovaného (nebo též aktivního) procesu.

Její deklarace je podle normy jazyka následující:


procedure cancel (x);


ref (process) x;


if x == current then

passivate


else

inspect x do

if EVENT =/= none then

begin


EVENT.out;



EVENT :- none

end
Je-li procedura cancel použita se skutečným parametrem, který odkazuje na právě aktivní exemplář třídy process, je její volání shodné s voláním procedury passivate.

V ostatních případech odstraní volání procedury cancel plán aktivace exempláře třídy process, který je dán jejím skutečným parametrem, z kalendáře událostí (existuje-li takový plán vůbec). - Viz obrázek.

((15.05.03.))

15.08. AKTIVUJÍCÍ PŘÍKAZY.

Jak bylo výše řečeno, rozumíme aktivujícími příkazy ty příkazy, které zařazují plány provedení exemplářů třídy process do simulačního kalendáře událostí.

15.08.01. Procedura activate.

Tvorbu exemplářů třídy EVENTNOTICE a jejich zařazování do kalendáře událostí SQS provádí procedura ACTIVATE. Tato procedura je pro uživatele nepřístupná a její vyvolávání se děje pomocí příkazů speciálního syntaktického tvaru.

Syntaktická definice procedury je podle normy jazyka dána následující deklarací:


procedure ACTIVATE (REAC, X, CODE, T, Y, PRIOR);


value CODE;


ref (process) X, Y;


boolean REAC, PRIOR;


text CODE;


real T;


inspect X do

if not TERMINATED then

begin


ref (process) Z;



ref (EVENTNOTICE) EV;



if REAC then EV :- EVENT



else


if EVENT =/= none then go to exit;



Z :- current;



if CODE eq "direct" then


direct:



begin



EVENT :- new EVENTNOTICE (time, X);




EVENT.precede (FIRSTEV)



end


else


if CODE eq "delay" then


begin



T := T + time;




go to at



end


else


if CODE eq "at" then


at:



begin



if T lt time then T := time;




if T eq time and PRIOR then go to direct;




EVENT :- new EVENTNOTICE (T, X);




EVENT.RANK (PRIOR)



end


else


if


(if Y == none then true else Y.EVENT == none)



then


EVENT :- none


else


begin



if X == Y then go to exit;




EVENT :- new EVENTNOTICE (Y.EVENT.EVTIME, X);




if CODE eq "before" then



EVENT.precede (Y.EVENT)




else



EVENT.follow (Y.EVENT)



end;



if EV =/= none then


begin



EV.out;




if SQS.empty then ERROR



end;



if Z =/= current then


resume (current);


exit:


end
15.08.02. Syntaktický tvar aktivujících příkazů.

Jak bylo již uvedeno výše, užívá se k vyvolání uživateli přímo nepřístupné procedury ACTIVATE několik speciálních příkazů, v nichž jsou obsaženy určité vyhrazené symboly jazyka (používané pouze k tomuto účelu).

15.08.02.01. Aktivátory.

Každý z aktivačních příkazů musí začínat tak zvaným aktivátorem.

Aktivátory rozeznáváme dva:


activate a reactivate
Obecný tvar aktivujícího příkazu je následující:


*AQ* nebo *AQU*

kde *A* je jeden z aktivátorů, *Q* je referenční výraz (označuje exemplář třídy process, kterého se aktivace týká - není-li tímto referenčním výrazem označen exemplář třídy process, dojde k chybě při výpočtu) a konečně *U* je určení doby aktivace, které může chybět.

Určení doby aktivace rozeznáváme časové a událostní.

Časové určení aktivace může mít jeden z následujících tvarů:


at *R*


at *R* prior

delay *R*


delay *R* prior
kde *R* je aritmetický výraz.

Událostní určení aktivace může mít tvar:


after *P*


before *P*

kde *P* je referenční výraz, který opět musí označovat exemplář třídy process, nemá-li dojít k chybě při výpočtu.

15.08.03. Význam jednotlivých aktivujících příkazů.

Nebude-li jinak řečeno, budeme v dalším předpokládat, že *?* je referenční výraz, který odkazuje na exemplář třídy process, který je v pasivním stavu.

15.08.03.01. Přímá aktivace.

Příkaz


activate *P*

znamená volání procedury ACTIVATE v následujícím tvaru:


ACTIVATE (false, *P*, "direct", 0, none, false);

a je tedy ekvivalentní provedení posloupnosti příkazů:


inspect *P* do

if not TERMINATED then

begin


if EVENT =/= none then go to exit;



EVENT :- new EVENTNOTICE (time, *P*);



EVENT.precede (FIRSTEV);



resume (current);


exit:


end
Uvedený příkaz tedy způsobí, že se na začátek kalendáře událostí zařadí plán provedení procesu *P* s časem rovným momentální hodnotě simulárního času. Exempláři *P* se předá řízení v rámci realizujícího kvaziparalelního procesu, plán provádění dosud aktivního exempláře třídy process zůstává zařazen v kalendáři událostí na druhém místě.

Je-li hodnotou výrazu *P* referenční konstanta none nebo je-li exemplář označený výrazem *P* ve stavu ukončeném, je příkaz ekvivalentní prázdnému příkazu (totéž platí i pro ostatní aktivující příkazy).

((15.06.01.)), ((15.06.02.)), ((15.06.03.))

15.08.03.02. At.

Příkaz


activate *P* at *S*

respektive


activate *P* at *S* prior
způsobí volání procedury ACTIVATE ve tvaru:


ACTIVATE (false, *P*, "at", *S*, none, false);

respektive


ACTIVATE (false, *P*, "at", *S*, none, true);

a znamená tedy v podstatě provedení následujících příkazů:


inspect *P* do

if not TERMINATED then

begin


if EVENT =/= none then go to exit;



if *S* lt time then *S* := time;



if *S* eq time and PRIOR then


begin



EVENT :- new EVENTNOTICE (time, *P*);




EVENT.precede (FIRSTEV)



end


else


begin



EVENT :- new EVENTNOTICE (*S*, *P*);




EVENT.RANK (PRIOR)



end;



resume (current);


exit:


end
Uvedený příkaz tedy způsobí, že se do kalendáře událostí zařadí plán aktivace procesu *P* s časem rovným maximu z hodnoty aritmetického výrazu *S* a hodnoty simulárního času.

Příkaz v kratší formě způsobí zařazení tohoto plánu za všechny plány se stejným časem, příkaz v delší formě naopak před všechny plány se stejným časem aktivace.

((15.06.04.)), ((15.06.05.)), ((15.06.06.))

15.08.03.03. Delay.

Příkaz


activate *P* delay *S*

respektive


activate *P* delay *S* prior
kde *S* je aritmetický výraz, způsobí volání procedury ACTIVATE s těmito parametry:


ACTIVATE (false, *P*, "delay", *S*, none, false);

respektive


ACTIVATE (false, *P*, "delay", *S*, none, true);

a je ekvivalentní příkazu:


activate *P* at time + *S*

respektive


activate *P* at time + *S* prior
((15.06.07.)), ((15.06.08.))

15.08.03.04. Before a after.

Příkaz


activate *P* before *Q*

respektive


activate *P* after *Q*

kde *?* je referenční výraz označující exemplář třídy process, způsobí volání procedury ACTIVATE ve tvaru:


ACTIVATE (false, *P*, "before", 0, *Q*, false);

respektive


ACTIVATE (false, *P*, "after", 0, *Q*, false);

a znamená provedení následujících příkazů:


inspect *P* do

if not TERMINATED then

begin


ref (process) Z;



if EVENT =/= none then go to exit;



Z :- current;



if


(if *Q* == none then true else *Q*.EVENT == none)



then


EVENT :- none


else


begin



EVENT :- new EVENTNOTICE (*Q*.EVENT.EVTIME, *P*);




if CODE eq "before" then



EVENT.precede (*Q*.EVENT)




else



EVENT.follow (*Q*.EVENT)



end;



if Z =/= current then


resume (current);


exit:


end
Prvý příkaz zařadí plán procesu *P* před plán provedení procesu *Q* se shodným časem provedení,

((15.06.09.))

druhý příkaz zařadí podobně plán provedení procesu *P* za plán provedení procesu *Q* se stejným časem provedení.

((15.06.10.))

Je-li hodnotou výrazu *Q* referenční konstanta none nebo je-li proces označený výrazem *Q* ve stavu pasivním nebo ukončeném, je příkaz ekvivalentní prázdnému příkazu.

Souhrnně ukazuje funkci uvedených aktivujících příkazů následující obrázek:

15.08.04. Aktivátor reactivate.

Použije-li se pro pasivní proces *P* ve výše uvedených příkazech místo aktivátoru activate aktivátor reactivate, smysl příkazů se nezmění.

Je-li ale výrazem *P* označen exemplář třídy process, který je ve stavu aktivním nebo suspendovaném, pak (jak bylo výše uvedeno) jsou příkazy:


activate *P* ~

ekvivalentní prázdnému příkazu.

Naopak příkazy tvaru


reactivate *P* ~

zruší dosavadní plán provádění procesu *P* a pak provedou akce popsané výše u příslušného příkazu s aktivátorem activate.

Přesněji:

15.08.04.01.

Příkaz


reactivate *P*

znamená volání procedury ACTIVATE se skutečnými parametry:


ACTIVATE (true, *P*, "direct", 0, none, false);

a je ekvivalentní příkazům:


inspect *P* do

if not TERMINATED then

begin


ref (process) Z;



ref (EVENTNOTICE) EV;



Z :- current;



EV :- EVENT;



EVENT :- new EVENTNOTICE (time, *P*);



EVENT.precede (FIRSTEV);



if EV =/= none then


EV.out;



if Z =/= current then


resume (current)


end
15.08.04.02.

Příkaz


reactivate *P* at *S*

respektive


reactivate *P* at *S* prior
znamená volání procedury ACTIVATE ve tvaru:


ACTIVATE (true, *P*, "at", *S*, none, false);

respektive


ACTIVATE (true, *P*, "at", *S*, none, true);

a to znamená provedení příkazů:


inspect *P* do

if not TERMINATED then

begin


ref (process) Z;



ref (EVENTNOTICE) EV;



Z :- current;



EV :- EVENT;



if *S* lt time then *S* := time;



EVENT :- new EVENTNOTICE (*S*, *P*);



if *S* eq time and PRIOR then


EVENT.precede (FIRSTEV)



else


EVENT.RANK (PRIOR);



if EV =/= none then


EV.out;



if Z =/= current then


resume (current)


end
15.08.04.03.

Konečně příkaz


reactivate *P* before *Q*

respektive


reactivate *P* after *Q*

znamená volání procedury ACTIVATE:


ACTIVATE (true, *P*, "before", 0, *Q*, false);

respektive


ACTIVATE (true, *P*, "after", 0, *Q*, false);

což znamená provedení příkazů:


inspect *P* do

if not TERMINATED then

begin


if *P* == *Q* then go to exit;



begin



ref (process) Z;




ref (EVENTNOTICE) EV;




Z :- current;




EV :- EVENT;




if



(if *Q* == none then true else *Q*.EVENT == none)




then



EVENT :- none



else



begin




EVENT :- new EVENTNOTICE (*Q*.EVENT.EVTIME, *P*);





if CODE eq "before" then




EVENT.precede (*Q*.EVENT)





else




EVENT.follow (*Q*.EVENT)




end;




if EV =/= none then



begin




EV.out;





if SQS.empty then ERROR




end;




if Z =/= current then



resume (current)



end;


exit:


end
Poznamenejme, že je-li exemplář třídy process, který je hodnotou výrazu *Q*, ve stavu ukončeném nebo pasivním, k zařazení plánu provedení procesu *P* do kalendáře nedojde, byl-li dále původní plán provádění procesu *P* poslední položkou kalendáře SQS, která nyní byla odstraněna, dojde k chybě při výpočtu.

((15.06.11.)), ((15.06.12.)), ((15.06.13.)), ((15.06.14.)), ((15.06.15.)), ((15.06.16.)), ((15.06.17.)), ((15.06.18.))

Následující obrázek schematicky ukazuje význam aktivujících příkazů s aktivátorem reactivate:

15.08.05. Parametry procedury activate.

Pro přehlednost uvedeme ještě tabulku skutečných parametrů procedury ACTIVATE při jejím volání v jednotlivých případech aktivujících příkazů.

příkaz                                       parametry

                             REAC     X    CODE        T    Y       PRIOR

activate p                   false    p    "direct"    0    none    false
activate p at s              false    p    "at"        s    none    false
activate p at s prior        false    p    "at"        s    none    true
activate p delay s           false    p    "delay"     s    none    false
activate p delay s prior     false    p    "delay"     s    none    true
activate p before q          false    p    "before"    0    q       false
activate d after q           false    p    "after"     0    q       false
reactivate p                 true     p    "direct"    0    none    false
reactivate p at s            true     p    "at"        s    none    false
reactivate p at s prior      true     p    "at"        s    none    true
reactivate p delay s         true     p    "delay"     s    none    false
reactivate p delay s prior   true     p    "delay"     s    none    true
reactivate p before q        true     p    "before"    0    q       false
reactivate p after q         true     p    "after"     0    q       false
15.09. VÝPOČET ČASOVÉHO INTEGRÁLU.

Poslední standardní procedura třídy simulation - procedura accum - slouží pro výpočet integrálu podle simulárního času.

Podle normy jazyka je popsána deklarací:


procedure accum (int, last, fce, delta);


name int, last, fce;


real int, last, fce, delta;


begin


int := int + fce * (time - last);



last := time;



fce := fce + delta


end
Procedura accum volaná při každé změně veličiny fce o hodnotu danou výrazem delta počítá sloupcový integrál hodnot této veličiny a jeho hodnotu ukládá do proměnné int. Proměnná last slouží k uložení hodnoty simulárního času při posledním volání procedury.

